

Estruturas Condicionais

Prof. Francisco Rapchan
www.geocities.com/chicorapchan
rapchan@terra.com.br

Introdução

Um algoritmo deve ter alguma forma de poder escolher entre diversos caminhos, aquele que irá seguir. A estrutura condicional permite esta escolha.

Exemplo 1. Faça um algoritmo que leia dois números e mostre o maior.

```
ALGORITMO "Mostra o maior"
  Var
    numero_um, numero_dois: numerico
INICIO
  Leia (numero_um)
  Leia (numero_dois)

  Se numero_um > numero_dois entao
    escreva (numero_um)
  senao
    escreva (numero_dois)
  fimse

FIMALGORITMO
```

Neste algoritmo precisamos fazer uma **decisão**: precisamos decidir qual dos dois números digitados pelo usuário é o maior. Para isso usamos uma “**Estrutura de Desvio Condicional**” chamada de **SE**. A estrutura **se** permite que o computador decida ir por um caminho ou por outro conforme o resultado de uma condição. No exemplo acima, se **numero_um** tiver um valor maior que **numero_dois**, o programa vai mostrar o valor de **numero_um**. Senão, ou seja, se o valor de **numero_um** não for maior que o de **numero_dois**, este último é que será mostrado.

A expressão **fimse** indica o fim da Estrutura SE.

Exemplo 2. Faça um algoritmo que leia um número e mostre se ele é positivo, negativo ou zero.

```
algoritmo "Mostra se é zero, positivo ou negativo"
  var
    numero: numerico
inicio
  Leia (numero)
  Se numero = 0 entao
    escreva ("É zero")
  senao
    Se numero > 0 entao
      escreva ("É positivo")
    senao
      escreva ("É negativo")
    fimse
  fimse
fimalgoritmo
```

Exemplo 3. Faça um algoritmo que leia dois números e mostre se são iguais. Se não forem iguais, mostre o maior.

```
ALGORITMO "Mostra o maior"
VAR
    numero_um, numero_dois: numerico

INICIO
    Leia (numero_um)
    Leia (numero_dois)

    Se numero_um = numero_dois entao
        escreva ("São iguais")
    senao

        Se numero_um > numero_dois entao
            escreva (numero_um)
        senao
            escreva (numero_dois)
        fimse
    fimse
FIMALGORITMO
```

Neste algoritmo precisamos tomar duas decisões: se os números forem iguais, apenas mostrar que são iguais. Mas se forem diferentes, vamos ter que decidir qual é o maior e mostrá-lo. Observe que temos uma estrutura *se* dentro da outra. Dizemos que temos **duas Estruturas SE aninhadas** ou um **ninho de Estruturas SE**.

Exemplo 4. Explique o que faz o algoritmo abaixo.

```
ALGORITMO "O que faz?"
VAR
    a,l,m,n : real

INICIO
    leia (l,m,n)

    se (l > m) ou (l > n) entao
        se (m < n) entao
            a <- l
            l <- m
            m <- a
        senao
            a <- l
            l <- n
            n <- a
        fimse
    fimse

    se (m > n) entao
        a <- m
        m <- n
        n <- a
    fimse

    escreva (l,m,n)

fimalgoritmo
```

Para facilitar, faça uma tabela de verificação para as variáveis.

a	l	m	n

Exemplo 5. Faça um algoritmo que leia três números e mostre o maior.

```
algoritmo "Maior de 3 números"
var
    n1, n2, n3, maior: real
inicio
    leia (n1,n2,n3)

    maior <- n1

    se n2 > maior entao
        maior <- n2
    fimse

    se n3 > maior entao
        maior <- n3
    fimse

    escreva (maior)

fimalgoritmo
```

Observe o comportamento da variável **maior**. Veja como ela troca de valor ao longo do fluxo para conter sempre o maior valor dentre todos os que foram lidos.

Os exemplos abaixo usam a estrutura condicional com operadores lógicos **E** e **OU**

Exemplo 6. Leia uma nota e mostre o conceito equivalente. Suponha a seguinte tabela de conversão.

Nota	Conceito
de 0.0 até 5.0	D
Acima de 5.0 até 7.0	C
Acima de 7.0 até 9.0	B
Acima de 9.0 até 10.0	A

```
algoritmo "Conceitos"
var
    nota: real
inicio
    leia (nota)

    se (nota >= 0) e (nota <= 5) entao
        escreva ("Conceito D")
    fimse
    se (nota > 5) e (nota <= 7) entao
        escreva ("Conceito C")
    fimse
    se (nota > 7) e (nota <= 9) entao
        escreva ("Conceito B")
    fimse
    se (nota > 9) e (nota <= 10) entao
        escreva ("Conceito A")
    fimse

fimalgoritmo
```

Exemplo 7. Faça um algoritmo que leia 3 números e diga se podem ser lados de um triângulo.

Observe: A, B e C são lados de um triângulo se: $A < (B+C)$ e $B < (C+A)$ e $C < (A+B)$.

```
algoritmo "Lados do triângulo"
var
  A, B, C: real
inicio
  leia (A,B,C)
  se (A<(B+C)) e (B<(C+A)) e (C<(A+B)) entao
    escreva ("São lados de um triângulo")
  senao
    escreva ("Não são lados de um triângulo")
  fimse
fimalgoritmo
```

Observe neste algoritmo o uso do operador lógico e

Exemplo 8. Faça um algoritmo que leia o valor dos lados de um triângulo e diga se é equilátero, isósceles ou escaleno.

- Equilátero se: $A = B = C$.
- Isósceles se: $A = B$ ou $B = C$ ou $A = C$
- Escaleno se: $A \neq B \neq C$

```
algoritmo "Lados do triângulo"
var
  A, B, C: real
inicio
  leia (A,B,C)
  se (A<(B+C)) e (B<(C+A)) e (C<(A+B)) entao
    se (A = B) e (B = C) entao
      escreva ("Triângulo equilátero")
    senao
      se (A = B) ou (A = C) ou (C = B) entao
        escreva ("Triângulo isósceles")
      senao
        escreva ("Triângulo escaleno")
    fimse
  fimse
senao
  escreva ("Não são lados de um triângulo")
fimse
fimalgoritmo
```

Observe neste algoritmo o uso dos operadores lógicos e e ou.

Exemplo 9. Faça um algoritmo que leia a média e a frequência de um aluno e mostre se ele foi aprovado. Os critérios de aprovação são:

- Se a média for maior ou igual a 7.0 e a frequência maior que 75 então aluno está aprovado.
- Se a média for maior que 9.0 o aluno é aprovado com qualquer frequência.
- Se a média for maior que 5.0 e frequência maior ou igual a 75 então o aluno fica para recuperação.

```
ALGORITMO "Resultado acadêmico"
  var media, frequencia: Numerico
INICIO
  Leia (media, frequencia)

  se (media >= 7) e (frequencia >= 75) entao
    escreva ("Aluno aprovado")
  senao

    se media >= 9 entao
      escreva ("Aluno aprovado")
    senao
      se (media >= 5) e (frequencia >= 75) entao
        escreva ("Aluno de recuperação")
      senao
        escreva ("Aluno reprovado")
      fimse
    fimse

  fimse

FINALGORITMO
```

Exemplo 10. Faça um algoritmo que leia o número do mês e informe o número de dias que ele tem.

```
algoritmo "Meses do ano"
var mes : inteiro

inicio
  leia (mes)

  escolha mes
    caso 1,3,5,7,8,10,12
      escreva ("Mês com 31 dias")
    caso 4,6,9,11
      escreva ("Mês com 30 dias")
    caso 2
      escreva ("Mês com 28 ou 29 dias")
    outrocaso
      escreva ("Não é um mês válido")
  fimsecolha

finalgoritmo
```

Exemplo 11. Faça um algoritmo que leia o nome do mês e informe o número de dias que ele tem.

```
algoritmo "Meses do ano"
var mes : caracter

inicio
    leia (mes)

    escolha mes
        caso "janeiro", "março", "maio", "julho", "agosto", "outubro", "dezembro"
            escreva ("Mês com 31 dias")
        caso "abril", "junho", "setembro", "novembro"
            escreva ("Mês com 30 dias")
        caso "fevereiro"
            escreva ("Mês com 28 ou 29 dias")
        outrocaso
            escreva ("Não é um mês válido")
    fimescolha

fimalgoritmo
```